```
 1: (* $Id: absyn.mli,v 1.9 2019-01-24 13:15:38-08 - - $ *)
 2:
 3: (*
 4: * Abstract syntax definitions for SB.
 5: *)
 6:
 7: type linenr    = int
 8: type ident     = string
 9: type label     = string
10: type number    = float
11: type oper      = string
12:
13: and  printable = Printexpr of expr
14:                | String of string
15:
16: and  memref    = Arrayref of ident * expr
17:                | Variable of ident
18:
19: and  expr      = Number of number
20:                | Memref of memref
21:                | Unary of oper * expr
22:                | Binary of oper * expr * expr
23:
24: type stmt      = Dim of ident * expr
25:                | Let of memref * expr
26:                | Goto of label
27:                | If of expr * label
28:                | Print of printable list
29:                | Input of memref list
30:
31: type progline  = linenr * label option * stmt option
32:
33: type program   = progline list
34:
```

```
 1: (* $Id: etc.mli,v 1.1 2019-01-18 11:49:38-08 - - $ *)
 2:
 3: (*
 4: * Main program and system access.
 5: *)
 6:
 7: val warn : string list -> unit
 8:
 9: val die : string list -> unit
10:
11: val syntax_error : Lexing.position -> string list -> unit
12:
13: val usage_exit : string list -> unit
14:
15: val read_number : unit -> float
16:
```

```
 1: (* $Id: etc.ml,v 1.1 2019-01-18 11:49:38-08 - - $ *)
 2:
 3: let execname = Filename.basename Sys.argv.(0)
 4:
 5: let exit_status_ref = ref 0
 6:
 7: let quit () =
 8:     if !Sys.interactive
 9:     then Printf.printf "Quit: exit %d\n%!" !exit_status_ref
10:     else exit !exit_status_ref
11:
12: let eprint_list message =
13:     (exit_status_ref := 1;
14:      flush_all ();
15:      List.iter prerr_string message;
16:      prerr_newline ();
17:      flush_all ())
18:
19: let warn message = eprint_list (execname :: ": " :: message)
20:
21: let die message = (warn message; quit ())
22:
23: let syntax_error position message =
24:     warn (position.Lexing.pos_fname :: ": "
25:             :: string_of_int position.Lexing.pos_lnum :: ": "
26:             :: message)
27:
28: let usage_exit message =
29:     (eprint_list ("Usage: " :: execname :: " " :: message); quit ())
30:
31: let buffer : string list ref = ref []
32:
33: let rec read_number () = match !buffer with
34:     | head::tail -> (buffer := tail;
35:                      try float_of_string head
36:                      with Failure _ -> nan)
37:     | [] -> let line = input_line stdin
38:             in (buffer := Str.split (Str.regexp "[ \\t]+") line;
39:                 read_number ())
40:
```

```
 1: (* Generated file: DO NOT EDIT *)
 2: (* ocamlc -i tables.ml *)
 3: (* Generated Thu Jan 24 18:32:49 PST 2019 *)
 4:
 5: type variable_table_t = (string, float) Hashtbl.t
 6: val variable_table : variable_table_t
 7:
 8: type array_table_t = (string, float array) Hashtbl.t
 9: val array_table : array_table_t
10:
11: type unary_fn_table_t = (string, float -> float) Hashtbl.t
12: val unary_fn_table : unary_fn_table_t
13:
14: type binary_fn_table_t = (string, float -> float -> float) Hashtbl.t
15: val binary_fn_table : binary_fn_table_t
16:
17: type label_table_t = (string, Absyn.program) Hashtbl.t
18: val label_table : label_table_t
19: val init_label_table : Absyn.program -> unit
20: val dump_label_table : unit -> unit
```

```
 1: (* $Id: tables.ml,v 1.2 2019-01-24 18:32:49-08 - - $ *)
 2:
 3: type variable_table_t = (string, float) Hashtbl.t
 4: let variable_table : variable_table_t = Hashtbl.create 16
 5: let _ = List.map (fun (label, value) ->
 6:                   Hashtbl.add variable_table label value)
 7:                  ["e"  , exp 1.0;
 8:                   "eof", 0.0;
 9:                   "pi" , acos ~-.1.0;
10:                   "nan", nan]
11:
12: type array_table_t = (string, float array) Hashtbl.t
13: let array_table : array_table_t = Hashtbl.create 16
14:
15: type unary_fn_table_t = (string, float -> float) Hashtbl.t
16: let unary_fn_table : unary_fn_table_t = Hashtbl.create 16
17: let _ = List.map (fun (label, value) ->
18:                   Hashtbl.add unary_fn_table label value)
19:                  ["+"    , (~+.);
20:                   "-"    , (~-.);
21:                   "abs"  , abs_float;
22:                   "acos" , acos;
23:                   "asin" , asin;
24:                   "atan" , atan;
25:                   "ceil" , ceil;
26:                   "cos"  , cos;
27:                   "exp"  , exp;
28:                   "floor", floor;
29:                   "log"  , log;
30:                   "log10", log10;
31:                   "log2" , (fun x -> log x /. log 2.0);
32:                   "round", (fun x -> floor (x +. 0.5));
33:                   "sin"  , sin;
34:                   "sqrt" , sqrt;
35:                   "tan"  , tan]
36:
37: type binary_fn_table_t = (string, float -> float -> float) Hashtbl.t
38: let binary_fn_table : binary_fn_table_t = Hashtbl.create 16
39: let _ = List.map (fun (label, value) ->
40:                   Hashtbl.add binary_fn_table label value)
41:                  ["+", (+.);
42:                   "-", (-.);
43:                   "*", ( *.);
44:                   "/", (/.);
45:                   "%", mod_float;
46:                   "^", ( ** )]
47:
```

```
48:
49: type label_table_t = (string, Absyn.program) Hashtbl.t
50: let label_table : label_table_t = Hashtbl.create 16
51:
52: let rec init_label_table program = match program with
53:     | [] -> ()
54:     | (_, Some label, _)::rest ->
55:                 (Hashtbl.add label_table label program;
56:                  init_label_table rest)
57:     | _::rest -> init_label_table rest
58:
59: let dump_label_table () =
60:     let dump key value = match value with
61:         | [] -> ()
62:         | (line, _, _)::_ ->
63:           Printf.fprintf stderr "label \"%s\": line %d\n%!" key line
64:     in Hashtbl.iter dump label_table
65:
```

```
1: (* $Id: interp.mli,v 1.1 2019-01-18 11:49:38-08 - - $ *)
2:
3: (*
4: * Interpreter for Silly Basic
5: *)
6:
7: val interpret_program : Absyn.program -> unit
8:
```

```
 1: (* $Id: interp.ml,v 1.6 2019-01-24 13:15:38-08 - - $ *)
 2:
 3: exception Unimplemented of string
 4: let unimpl reason = raise (Unimplemented reason)
 5:
 6: let rec eval_expr (expr : Absyn.expr) : float = match expr with
 7:     | Absyn.Number number -> number
 8:     | Absyn.Memref memref -> unimpl "eval_expr Memref"
 9:     | Absyn.Unary (oper, expr) -> unimpl "eval_expr Unary"
10:     | Absyn.Binary (oper, expr1, expr2) -> unimpl "eval_expr Binary"
11:
12: let interp_print (print_list : Absyn.printable list) =
13:     let print_item item =
14:         (print_char ' ';
15:          match item with
16:          | Absyn.String string ->
17:            let regex = Str.regexp "\"\\(.*\\)\""
18:            in print_string (Str.replace_first regex "\\1" string)
19:          | Absyn.Printexpr expr ->
20:            print_float (eval_expr expr))
21:     in (List.iter print_item print_list; print_newline ())
22:
23: let interp_input (memref_list : Absyn.memref list) =
24:     let input_number (memref : Absyn.memref) =
25:         try  let number = Etc.read_number ()
26:              in (print_float number; print_newline ())
27:         with End_of_file ->
28:              (print_string "End_of_file"; print_newline ())
29:     in List.iter input_number memref_list
30:
31: let interp_stmt (stmt : Absyn.stmt) = match stmt with
32:     | Absyn.Dim (ident, expr) -> unimpl "Dim (ident, expr)"
33:     | Absyn.Let (memref, expr) -> unimpl "Let (memref, expr)"
34:     | Absyn.Goto labsl -> unimpl "Goto labsl"
35:     | Absyn.If (expr, label) -> unimpl "If (expr, label)"
36:     | Absyn.Print print_list -> interp_print print_list
37:     | Absyn.Input memref_list -> interp_input memref_list
38:
39: let rec interpret (program : Absyn.program) = match program with
40:     | [] -> ()
41:     | firstline::otherlines -> match firstline with
42:       | _, _, None -> interpret otherlines
43:       | _, _, Some stmt -> (interp_stmt stmt; interpret otherlines)
44:
45: let interpret_program program =
46:     (Tables.init_label_table program;
47:      interpret program)
48:
```

```
 1: (* $Id: main.ml,v 1.1 2019-01-18 11:49:38-08 - - $ *)
 2:
 3: (*
 4: * Main program reads a file and prints to stdout.
 5: *)
 6:
 7: let interpret_source filename =
 8:     try (let sourcefile =
 9:              if filename = "-"
10:              then stdin
11:              else open_in filename in
12:          let lexbuf = Lexing.from_channel sourcefile in
13:          let abstract_syntax = Parser.program Scanner.token lexbuf in
14:          Interp.interpret_program abstract_syntax)
15:     with Sys_error (string) -> Etc.die [string]
16:
17: let _ = if !Sys.interactive
18:         then ()
19:         else match Array.length Sys.argv with
20:              | 1 -> interpret_source "-"
21:              | 2 -> interpret_source Sys.argv.(1)
22:              | _ -> Etc.usage_exit ["[filename.sb]"]
23:
```

```
 1: /* $Id: parser.mly,v 1.4 2019-01-24 13:15:38-08 - - $ */
 2:
 3: %{
 4:
 5: let linenr () = (symbol_start_pos ()).Lexing.pos_lnum
 6:
 7: let syntax () = Etc.syntax_error (symbol_start_pos ()) ["syntax error"]
 8:
 9: %}
10:
11: %token <string> RELOP EQUAL ADDOP MULOP POWOP
12: %token <string> IDENT NUMBER STRING
13: %token COLON COMMA LPAR RPAR LSUB RSUB EOL EOF
14: %token DIM LET GOTO IF PRINT INPUT
15:
16: %type <Absyn.program> program
17:
18: %start program
19:
20: %%
21:
22: program    : stmt_list EOF             {List.rev $1}
23:
24: stmt_list  : stmt_list stmt EOL        {$2::$1}
25:            | stmt_list error EOL       {syntax (); $1}
26:            |                           {[]}
27:
28: stmt       : label action              {(linenr (), Some $1, Some $2)}
29:            | action                    {(linenr (), None, Some $1)}
30:            | label                     {(linenr (), Some $1, None)}
31:            |                           {(linenr (), None, None)}
32:
33: label      : IDENT COLON               {$1}
34:
35: action     : DIM IDENT LSUB expr RSUB {Absyn.Dim ($2, $4)}
36:            | LET memref EQUAL expr     {Absyn.Let ($2, $4)}
37:            | GOTO IDENT                {Absyn.Goto $2}
38:            | IF relexpr GOTO IDENT     {Absyn.If ($2, $4)}
39:            | PRINT print_list          {Absyn.Print $2}
40:            | PRINT                     {Absyn.Print ([])}
41:            | INPUT input_list          {Absyn.Input $2}
42:
43: print_list : print COMMA print_list    {$1::$3}
44:            | print                     {[$1]}
45:
46: print      : expr                      {Absyn.Printexpr $1}
47:            | STRING                    {Absyn.String $1}
48:
49: input_list : memref COMMA input_list   {$1::$3}
50:            | memref                    {[$1]}
51:
```

```
   52:
   53: memref     : IDENT                    {Absyn.Variable $1}
   54:            | IDENT LSUB expr RSUB     {Absyn.Arrayref ($1, $3)}
   55:
   56: relexpr    : expr RELOP expr          {Absyn.Binary ($2, $1, $3)}
   57:            | expr EQUAL expr          {Absyn.Binary ($2, $1, $3)}
   58:
   59: expr       : expr ADDOP term          {Absyn.Binary ($2, $1, $3)}
   60:            | term                     {$1}
   61:
   62: term       : term MULOP factor        {Absyn.Binary ($2, $1, $3)}
   63:            | factor                   {$1}
   64:
   65: factor     : primary POWOP factor     {Absyn.Binary ($2, $1, $3)}
   66:            | primary                  {$1}
   67:
   68: primary    : LPAR expr RPAR           {$2}
   69:            | ADDOP primary            {Absyn.Unary ($1, $2)}
   70:            | NUMBER                   {Absyn.Number (float_of_string $1)
}
   71:            | memref                   {Absyn.Memref $1}
   72:            | IDENT LPAR expr RPAR     {Absyn.Unary ($1, $3)}
   73:
```

```
 1: (* $Id: scanner.mll,v 1.1 2019-01-18 11:49:38-08 - - $ *)
 2:
 3: {
 4:
 5: let lexerror lexbuf =
 6:     Etc.syntax_error (Lexing.lexeme_start_p lexbuf)
 7:             ["invalid character `" ^ (Lexing.lexeme lexbuf) ^ "'"]
 8:
 9: let newline lexbuf =
10:     let incr pos =
11:         {pos with Lexing.pos_lnum = pos.Lexing.pos_lnum + 1;
12:                   Lexing.pos_bol = pos.Lexing.pos_cnum}
13:     in  (lexbuf.Lexing.lex_start_p <- incr lexbuf.Lexing.lex_start_p;
14:          lexbuf.Lexing.lex_curr_p <- incr lexbuf.Lexing.lex_curr_p)
15:
16: let lexeme = Lexing.lexeme
17:
18: }
19:
20: let letter          = ['a'-'z' 'A'-'Z' '_']
21: let digit           = ['0'-'9']
22: let fraction        = (digit+ '.'? digit* | '.' digit+)
23: let exponent        = (['E' 'e'] ['+' '-']? digit+)
24:
25: let comment         = ('#' [^'\n']*)
26: let ident           = (letter (letter | digit)*)
27: let number          = (fraction exponent?)
28: let string          = '"' [^'\n' '"']* '"'
29:
```

```
30:
31: rule token     = parse
32:      | eof            { Parser.EOF }
33:      | [' ' '\t']     { token lexbuf }
34:      | comment        { token lexbuf }
35:      | "\n"           { newline lexbuf; Parser.EOL }
36:      | ":"            { Parser.COLON }
37:      | ","            { Parser.COMMA }
38:      | "("            { Parser.LPAR }
39:      | ")"            { Parser.RPAR }
40:      | "["            { Parser.LSUB }
41:      | "]"            { Parser.RSUB }
42:      | "="            { Parser.EQUAL (lexeme lexbuf) }
43:      | "<>"           { Parser.RELOP (lexeme lexbuf) }
44:      | "<"            { Parser.RELOP (lexeme lexbuf) }
45:      | "<="           { Parser.RELOP (lexeme lexbuf) }
46:      | ">"            { Parser.RELOP (lexeme lexbuf) }
47:      | ">="           { Parser.RELOP (lexeme lexbuf) }
48:      | "+"            { Parser.ADDOP (lexeme lexbuf) }
49:      | "-"            { Parser.ADDOP (lexeme lexbuf) }
50:      | "*"            { Parser.MULOP (lexeme lexbuf) }
51:      | "/"            { Parser.MULOP (lexeme lexbuf) }
52:      | "%"            { Parser.MULOP (lexeme lexbuf) }
53:      | "^"            { Parser.POWOP (lexeme lexbuf) }
54:      | "dim"          { Parser.DIM }
55:      | "goto"         { Parser.GOTO }
56:      | "if"           { Parser.IF }
57:      | "input"        { Parser.INPUT }
58:      | "let"          { Parser.LET }
59:      | "print"        { Parser.PRINT }
60:      | number         { Parser.NUMBER (lexeme lexbuf) }
61:      | string         { Parser.STRING (lexeme lexbuf) }
62:      | ident          { Parser.IDENT (lexeme lexbuf) }
63:      | _              { lexerror lexbuf; token lexbuf }
64:
```

```
# $Id: Makefile,v 1.14 2019-01-24 14:04:36-08 - - $

#
# General useful macros
#

MKFILE     = Makefile
MAKEFLAGS += --no-builtin-rules
DEPSFILE   = ${MKFILE}.deps
NOINCLUDE  = ci clean spotless
NEEDINCL   = ${filter ${NOINCLUDE}, ${MAKECMDGOALS}}
GMAKE      = ${MAKE} --no-print-directory

#
# File list macros
#

EXECBIN    = sbinterp
OBJCMO     = etc.cmo parser.cmo scanner.cmo \
             tables.cmo interp.cmo main.cmo
OBJCMI     = ${OBJCMO:.cmo=.cmi} absyn.cmi
OBJBIN     = ${OBJCMO:.cmo=.o}
MLSOURCE   = absyn.mli etc.mli etc.ml tables.mli tables.ml \
             interp.mli interp.ml main.ml
GENSOURCE  = tables.mli parser.mli parser.ml scanner.ml
GENFILES   = ${GENSOURCE} parser.output ${DEPSFILE}
OTHERFILES = ${MKFILE} ${DEPSFILE} using
ALLSOURCES = ${MLSOURCE} parser.mly scanner.mll ${OTHERFILES}
LISTING    = Listing.ps

#
# General targets
#

all : ${EXECBIN}

${EXECBIN} : ${OBJCMO}
        ocamlc str.cma ${OBJCMO} -o ${EXECBIN}

%.cmi : %.mli
        ocamlc -c $<

%.cmo : %.ml
        ocamlc -c $<

%.ml : %.mll
        ocamllex $<

%.mli %.ml : %.mly
        ocamlyacc -v $<

```

```
  52:
  53: GEN_TABLES_MLI = ocamlc -i tables.ml
  54: tables.mli : tables.ml absyn.cmi
  55:         ( echo "(* Generated file: DO NOT EDIT *)" \
  56:         ; echo "(* ${GEN_TABLES_MLI} *)" \
  57:         ; echo "(* Generated $$(date) *)" \
  58:         ; ${GEN_TABLES_MLI} | sed 's/^type/\n&/' \
  59:         ) >tables.mli
  60:
  61: #
  62: # Misc targets
  63: #
  64:
  65: clean :
  66:         - rm ${OBJCMI} ${OBJCMO} ${OBJBIN} tables.mli
  67:
  68: spotless : clean
  69:         - rm ${EXECBIN} ${GENFILES} ${LISTING} ${LISTING:.ps=.pdf}
  70:
  71: ci : ${ALLSOURCES}
  72:         cid + ${ALLSOURCES}
  73:
  74: GEN_OCAMLDEP = ocamldep ${MLSOURCE} ${GENSOURCE}
  75: GEN_FORMAT = perl -pe 's/^(.{1,72})\s+(.*)/$$1 \\\n\# $$2/'
  76: deps : ${MLSOURCE} ${GENSOURCE}
  77:         ( echo "# Generated file: DO NOT EDIT" \
  78:         ; echo "# ${GEN_OCAMLDEP}" | ${GEN_FORMAT} \
  79:         ; echo "# Generated $$(date)" \
  80:         ; ${GEN_OCAMLDEP} \
  81:         ) >${DEPSFILE}
  82:
  83: ${DEPSFILE} : tables.mli
  84:         @touch ${DEPSFILE}
  85:         ${GMAKE} deps
  86:
  87: lis : ${ALLSOURCES}
  88:         mkpspdf ${LISTING} ${ALLSOURCES}
  89:
  90: again :
  91:         ${GMAKE} spotless
  92:         ${GMAKE} deps
  93:         ${GMAKE} ci
  94:         ${GMAKE} all
  95:         ${GMAKE} lis
  96:
  97: ifeq "${NEEDINCL}" ""
  98: include ${DEPSFILE}
  99: endif
 100:
```

```
 1: # Generated file: DO NOT EDIT
 2: # ocamldep absyn.mli etc.mli etc.ml tables.mli tables.ml interp.mli \
 3: # interp.ml main.ml tables.mli parser.mli parser.ml scanner.ml
 4: # Generated Thu Jan 24 18:32:49 PST 2019
 5: absyn.cmi :
 6: etc.cmi :
 7: etc.cmo : etc.cmi
 8: etc.cmx : etc.cmi
 9: tables.cmi : absyn.cmi
10: tables.cmo : absyn.cmi tables.cmi
11: tables.cmx : absyn.cmi tables.cmi
12: interp.cmi : absyn.cmi
13: interp.cmo : tables.cmi etc.cmi absyn.cmi interp.cmi
14: interp.cmx : tables.cmx etc.cmx absyn.cmi interp.cmi
15: main.cmo : scanner.cmo parser.cmi interp.cmi etc.cmi
16: main.cmx : scanner.cmx parser.cmx interp.cmx etc.cmx
17: tables.cmi : absyn.cmi
18: parser.cmi : absyn.cmi
19: parser.cmo : etc.cmi absyn.cmi parser.cmi
20: parser.cmx : etc.cmx absyn.cmi parser.cmi
21: scanner.cmo : parser.cmi etc.cmi
22: scanner.cmx : parser.cmx etc.cmx
```

```
 1: (* $Id: using,v 1.2 2019-01-24 18:32:49-08 - - $ *)
 2:
 3: #load "str.cma";;
 4:
 5: #mod_use "absyn.mli";;
 6: #mod_use "etc.ml";;
 7:
 8: #mod_use "parser.ml";;
 9: #mod_use "scanner.ml";;
10:
11: #mod_use "tables.ml";;
12: #mod_use "interp.ml";;
13: #mod_use "main.ml";;
14:
15: open Interp;;
16: open Main;;
17:
```